```
(* HIABXS031 forward kinematics *)
(* Author: Goran Petrovic *)
(* Latest version date: 26. 01. 2023.*)

(* All the calculation data below are taken from the crane CAD files. *)
```

In[ ]:=
```
(* Constant auxiliary matrices *)
I3x3 = IdentityMatrix[3];
O3x3 = ConstantArray[0, {3, 3}];
I8x8 = IdentityMatrix[8];
O8x8 = ConstantArray[0, {8, 8}];
I2x2 = IdentityMatrix[2];
O2x2 = ConstantArray[0, {2, 2}];

(**************** Fixed geometric properties ****************)
(******************** Revolute segment 1 ********************)
(***************** Relevant lengths and angles **************)
L1 = 1.0071227324212;
L11 = 0.37610632101293;
lc1 = 0.7450;
x10 = 0.830000000000000;
α1 = 1.9515432018109;
α2 = 0.24548563270531;
β1 = 0.380746995657807;
β2 = 0.121961735475489;
(******************** Revolute segment 2 ********************)
(***************** Relevant lengths and angles **************)
L2 = 1.132185817991000;
L21 = 0.309011326653250;
lc2 = 0.7450;
x20 = 0.830000000000000;
α3 = 13.00842633118 * Pi / 180;
β3 = 0.122594643555178;
β4 = 0.288982033065996;
```

In[ ]:=
```
(************************************ RELEVANT
 POSITION VECTORS ************************************)
(* World frame {W} distance from pillar-base frame {Bp} expressed in {Bp} *)
WrWBc1 = {{0.185}, {-0.085}, {0}};
(* Pillar/Lift-link frame {B11} distance from pillar frame {Bc1} expressed in {Bc1} *)
Bc1rBc1B11 = {{L1}, {0}, {0}};
(* Lift-link frame {Tc1} distance from pillar/lift-
 link frame {B11} expressed in {B11} *)
B11rB11Tc1 = {{L11}, {0}, {0}};
(* Lift-link frame {Bc2} distance from lift-link frame {Tc1} expressed in {Tc1} *)
Tc1rTc1Bc2 = {{0.098495302558553}, {0.01838181329939}, {0}};
(* Lift-link/Tilt-link frame {B12} distance from lift-
 link frame {Bc2} expressed in {Bc2} *)
Bc2rBc2B12 = {{L2}, {0}, {0}};
(* Tilt-link frame {Tc2} distance from lift-
 link/tilt-link frame {B12} expressed in {B12} *)
B12rB12Tc2 = {{L21}, {0}, {0}};
(* Tilt-link/Extension-cylinder-
 1 frame {P22} distance from tilt-link frame {Tc2} expressed in {Tc2} *)
Tc2rTc2P22 = {{-0.048858374920227}, {0.36587359346146}, {0.035276938723102}};
(* Extension-piston-1 frame {B52} distance from tilt-
 link/extension-cylinder-1 frame {P22} expressed in {P22} *)
P22rP22B52 = {{0.041 + w3}, {0}, {0}};
(* Extension-piston-1/Extension-mass-
 1 frame {P32} distance from extension-piston-1 frame {B52} expressed in {B52} *)
B52rB52P32 = {{1.2}, {0}, {0}};
(* TCP frame {D22} distance from extension-piston-
 1/extension-mass-1 frame {P32} expressed in {P32} *)
P32rP32D22 = {{0.40499744871392}, {-0.13130517281345}, {-0.013385243295654}};
```

In[ ]:=
```
(************************************ RELEVANT
 ROTATION MATRICES ************************************)
(* Rotation matrix from pillar frame {Bc1} to world frame {W} *)
WRotBc1 = {{Cos[α1], -Sin[α1], 0}, {Sin[α1], Cos[α1], 0}, {0, 0, 1}};
(* Auxiliary angle q1. *)
q1 = -Pi / 2 - β1 - β2 + w1;
(* Auxiliary angle q2 *)
q2 = w2 - β3 - β4;
(* Rotation matrix from pillar/lift-link frame {B11} to pillar frame {Bc1} *)
Bc1RotB11 = {{Cos[q1], -Sin[q1], 0}, {Sin[q1], Cos[q1], 0}, {0, 0, 1}};
(* Rotation matrix from revolute-segment-
 1 cylinder frame {B31} to pillar frame {Bc1} *)
Bc1RotB31 = {{Cos[q11], -Sin[q11], 0}, {Sin[q11], Cos[q11], 0}, {0, 0, 1}};
(* Rotation matrix from revolute-segment-
 1 piston frame {B41} to revolute-segment-1 cylinder frame {B31} *)
B31RotB41 = IdentityMatrix[3];
(* Rotation matrix from lift-link frame {Tc1} to pillar/lift-link frame {B11} *)
B11RotTc1 = IdentityMatrix[3];
(* Rotation matrix from lift-link frame {Bc2} to lift-link frame {Tc1} *)
Tc1RotBc2 = {{Cos[α2], -Sin[α2], 0}, {Sin[α2], Cos[α2], 0}, {0, 0, 1}};
(* Rotation matrix from lift-link/tilt-link frame {B12} to lift-link frame {Bc2} *)
Bc2RotB12 = {{Cos[q2], -Sin[q2], 0}, {Sin[q2], Cos[q2], 0}, {0, 0, 1}};
(* Rotation matrix from revolute-segment-
 2 cylinder frame {B32} to lift-link frame {Bc2} *)
Bc2RotB32 = {{Cos[q21], -Sin[q21], 0}, {Sin[q21], Cos[q21], 0}, {0, 0, 1}};
(* Rotation matrix from revolute-segment-
 2 piston frame {B42} to revolute-segment-2 cylinder frame {B32} *)
B32RotB42 = IdentityMatrix[3];
(* Rotation matrix from tilt-link frame {Tc2} to lift-link/tilt-link frame {B12}  *)
B12RotTc2 = IdentityMatrix[3];
(* Rotation matrix from tilt-link/extension-
 cylinder-1 frame {P22} to tilt-link frame {Tc2} *)
Tc2RotP22 = {{0.95853460763586, -0.27664713597974, 0.0683942111482}, {0.28497614981675,
    0.93051946315663, -0.23004808784757}, {0, 0.23999977258511, 0.9707729441837}};
(* Rotation matrix from extension-piston-
 1 frame {B52} to extension-cylinder-1 frame {P22} *)
P22RotB52 = IdentityMatrix[3];
(* Rotation matrix from extension-piston-1/extension-
 mass-1 frame {P32} to extension-piston-1 frame {B52} *)
B52RotP32 = IdentityMatrix[3];
(* Rotation matrix from TCP to extension-piston-1/extension-mass-1 frame {P32} *)
P32RotD22 = {{1, 0, 0}, {0, 0.9707729441837, 0.23999977258512},
   {0, -0.23999977258512, 0.9707729441837}};
```

```
In[ ]:= (* Pillar/Lift-link frame {B11} position in world frame {W} *)
WrWB11 = Chop[FullSimplify[WrWBc1 + WRotBc1.Bc1rBc1B11]];
(* Rotation matrix from pillar/lift-link frame {B11} to world frame {W} *)
WRotB11 = Chop[FullSimplify[WRotBc1.Bc1RotB11]];

In[ ]:= WrWB11

Out[ ]= {{-0.189261}, {0.85}, {0}}

In[ ]:= (* Lift-link frame {Tc1} position in world frame {W} *)
WrWTc1 = Chop[FullSimplify[WrWB11 + WRotB11.B11rB11Tc1]];
(* Rotation matrix from lift-link frame {Tc1} to world frame {W} *)
WRotTc1 = Chop[FullSimplify[WRotB11.B11RotTc1]];
(* Lift-link frame {Bc2} position in world-frame {W} *)
WrWBc2 = Chop[FullSimplify[WrWTc1 + WRotTc1.Tc1rTc1Bc2]];
(* Rotation matrix from tilt-link frame {Bc2} to world frame {W} *)
WRotBc2 = Chop[FullSimplify[WRotTc1.Tc1RotBc2]];
(* Lift-link/Tilt-link frame {B12} position in world frame {W} *)
WrWB12 = Chop[FullSimplify[WrWBc2 + WRotBc2.Bc2rBc2B12]];
(* Rotation matrix from lift-link/tilt-link frame {B12} to world frame {W} *)
WRotB12 = Chop[FullSimplify[WRotBc2.Bc2RotB12]];

In[ ]:= WrWB12

Out[ ]= {{-0.189261 + 1.59687 Cos[w1] - 0.100002 Sin[w1]},
  {0.85 + 0.100002 Cos[w1] + 1.59687 Sin[w1]}, {0}}

In[ ]:= (* Rotation matrix from tilt-link frame {Tc2} to world frame {W} *)
WRotTc2 = Chop[FullSimplify[WRotB12.B12RotTc2]];
(* Rotation matrix from tilt-link/extension-
 cylinder-1 frame {P22} to world frame {W} *)
WRotP22 = Chop[FullSimplify[WRotTc2.Tc2RotP22]];
(* Tilt-link/Extension-cylinder-1 frame {P22} position in world frame {W} *)
WrWP22 = Chop[FullSimplify[WrWB12 + WRotB12.B12rB12Tc2 + WRotTc2.Tc2rTc2P22]];

In[ ]:= WrWP22

Out[ ]= {{-0.189261 + 1.59687 Cos[w1] +
    0.353374 Cos[w1 + w2] - 0.100002 Sin[w1] - 0.276893 Sin[w1 + w2]},
  {0.85 + 0.100002 Cos[w1] + 0.276893 Cos[w1 + w2] + 1.59687 Sin[w1] + 0.353374 Sin[w1 + w2]},
  {0.0352769}}

In[ ]:= (* Extension-piston-1 frame {B52} position in world frame {W} *)
WrWB52 = Chop[FullSimplify[WrWP22 + WRotP22.P22rP22B52]];
(* Rotation matrix from extension-piston-1 frame {B52} to world frame {W} *)
WRotB52 = Chop[FullSimplify[WRotP22.P22RotB52]];
```

In[ ]:=
```
WrWB52
```

Out[ ]=
```
{{-0.189261 + (0.394374 + 1. w3) Cos[w1 + w2] +
    (-0.100002 + (-0.0000380745 - 0.000928646 w3) Cos[w2]) Sin[w1] +
   Cos[w1] (1.59687 + (-0.0000380745 - 0.000928646 w3) Sin[w2]) - 0.276893 Sin[w1 + w2]},
 {0.85 + 0.100002 Cos[w1] + 0.276893 Cos[w1 + w2] + 1.59687 Sin[w1] + (0.041 + w3)
    (Sin[w1] (1. Cos[w2] - 0.000928646 Sin[w2]) + Cos[w1] (0.000928646 Cos[w2] + 1. Sin[w2])) +
   0.353374 Sin[w1 + w2]}, {0.0352769}}
```

In[ ]:=
```
(* Extension-piston-1/Extension-mass-1 frame {P32} position in world frame {W} *)
WrWP32 = Chop[FullSimplify[WrWB52 + WRotB52.B52rB52P32]];
(* Rotation matrix from extension-piston-
 1/extension-mass-1 frame {P32} to world frame {W} *)
WRotP32 = Chop[FullSimplify[WRotB52.B52RotP32]];
```

In[ ]:=
```
WrWP32
```

Out[ ]=
```
{{-0.189261 + (1.59437 + 1. w3) Cos[w1 + w2] +
    (-0.100002 + (-0.00115245 - 0.000928646 w3) Cos[w2]) Sin[w1] +
   Cos[w1] (1.59687 + (-0.00115245 - 0.000928646 w3) Sin[w2]) - 0.276893 Sin[w1 + w2]},
 {0.85 + Sin[w1] (1.59687 + (1.59437 + 1. w3) Cos[w2] + (-0.278046 - 0.000928646 w3) Sin[w2]) +
   Cos[w1]
    (0.100002 + (0.278046 + 0.000928646 w3) Cos[w2] + (1.59437 + 1. w3) Sin[w2])}, {0.0352769}}
```

In[ ]:=
```
(* Extension-mass-1/Extension-cylinder-
 2 frame {P22p} frame origin in world frame {W} *)
WrWD22 = Chop[FullSimplify[WrWP32 + WRotP32.P32rP32D22 ]];
(* Rotation matrix from extension-mass-
 1/extension-cylinder-2 frame {P22p} to world frame {W} *)
WRotD22 = Chop[FullSimplify[WRotP32.P32RotD22]];
```

In[ ]:=
```
WrWD22
```

Out[ ]=
```
{{-0.189261 + (1.99949 + 1. w3) Cos[w1 + w2] +
    (-0.100002 + (-0.00152855 - 0.000928646 w3) Cos[w2]) Sin[w1] +
   Cos[w1] (1.59687 + (-0.00152855 - 0.000928646 w3) Sin[w2]) - 0.152638 Sin[w1 + w2]},
 {0.85 + Sin[w1] (1.59687 + (1.99949 + 1. w3) Cos[w2] + (-0.154167 - 0.000928646 w3) Sin[w2]) +
   Cos[w1] (0.100002 + (0.154167 + 0.000928646 w3) Cos[w2] + (1.99949 + 1. w3) Sin[w2])},
 {-0.0092303}}
```

In[ ]:=
```
r[t_] := WrWD22 /. {w1 → w1[t], w2 → w2[t], w3 → w3[t]}
```

In[ ]:=
```
v[t_] := Chop[FullSimplify[D[r[t], {t, 1}]], 0.01]
```

```
In[•]:= v[t]
```

```
Out[•]= {{-0.100002 Cos[w1[t]] w1'[t] -
     1.59687 Sin[w1[t]] w1'[t] - 0.152638 Cos[w1[t] + w2[t]] (w1'[t] + w2'[t]) -
     1. Sin[w1[t] + w2[t]] (1.99949 + 1. w3[t]) (w1'[t] + w2'[t]) + 1. Cos[w1[t] + w2[t]] w3'[t]},
    {-Sin[w1[t]] (0.100002 + 0.154167 Cos[w2[t]] + Sin[w2[t]] (1.99949 + 1. w3[t])) w1'[t] +
     Cos[w1[t]] (1.59687 - 0.154167 Sin[w2[t]] + Cos[w2[t]] (1.99949 + 1. w3[t])) w1'[t] +
     Sin[w1[t]] ((-0.154167 Cos[w2[t]] - 1.99949 Sin[w2[t]] - 1. Sin[w2[t]] w3[t]) w2'[t] +
        1. Cos[w2[t]] w3'[t]) +
     Cos[w1[t]] ((1.99949 Cos[w2[t]] - 0.154167 Sin[w2[t]] + 1. Cos[w2[t]] w3[t]) w2'[t] +
        1. Sin[w2[t]] w3'[t])}, {0}}
```

```
In[•]:= V =
 D[(v[t] /. {w1[t] → w1, w1'[t] → dw1, w2[t] → w2, w2'[t] → dw2, w3[t] → w3, w3'[t] → dw3})]
```

```
Out[•]= {{-0.100002 dw1 Cos[w1] - 0.152638 (dw1 + dw2) Cos[w1 + w2] + 1. dw3 Cos[w1 + w2] -
     1.59687 dw1 Sin[w1] - 1. (dw1 + dw2) (1.99949 + 1. w3) Sin[w1 + w2]},
    {dw1 Cos[w1] (1.59687 + (1.99949 + 1. w3) Cos[w2] - 0.154167 Sin[w2]) +
     Cos[w1] (dw2 (1.99949 Cos[w2] + 1. w3 Cos[w2] - 0.154167 Sin[w2]) + 1. dw3 Sin[w2]) -
     dw1 Sin[w1] (0.100002 + 0.154167 Cos[w2] + (1.99949 + 1. w3) Sin[w2]) +
     Sin[w1] (1. dw3 Cos[w2] + dw2 (-0.154167 Cos[w2] - 1.99949 Sin[w2] - 1. w3 Sin[w2]))}, {0}}
```

```
In[•]:= J11 = FullSimplify[D[V[[1, 1]], {dw1, 1}]]
        J12 = FullSimplify[D[V[[1, 1]], {dw2, 1}]]
        J13 = FullSimplify[D[V[[1, 1]], {dw3, 1}]]
```

```
Out[•]= -0.100002 Cos[w1] - 0.152638 Cos[w1 + w2] - 1.59687 Sin[w1] + (-1.99949 - 1. w3) Sin[w1 + w2]
```

```
Out[•]= -0.152638 Cos[w1 + w2] + (-1.99949 - 1. w3) Sin[w1 + w2]
```

```
Out[•]= 1. Cos[w1 + w2]
```

```
In[•]:= J21 = FullSimplify[D[V[[2, 1]], {dw1, 1}]]
        J22 = FullSimplify[D[V[[2, 1]], {dw2, 1}]]
        J23 = FullSimplify[D[V[[2, 1]], {dw3, 1}]]
```

```
Out[•]= Cos[w1] (1.59687 + (1.99949 + 1. w3) Cos[w2] - 0.154167 Sin[w2]) +
 Sin[w1] (-0.100002 - 0.154167 Cos[w2] + (-1.99949 - 1. w3) Sin[w2])
```

```
Out[•]= Cos[w1] ((1.99949 + 1. w3) Cos[w2] - 0.154167 Sin[w2]) +
 Sin[w1] (-0.154167 Cos[w2] + (-1.99949 - 1. w3) Sin[w2])
```

```
Out[•]= 1. Cos[w2] Sin[w1] + 1. Cos[w1] Sin[w2]
```